Environ. Sci. Technol. 1994, 28, 898-902

Hypothesis Testing with Values below Detection Limit in Environmental Studies

Donald J. Slymen* **and Ann de Peyster**

Graduate School of Public Health, San Diego State University, San Diego, California 92182

Regina R. Donohoe

Department of Fisheries and Wildlife, Oregon State University, Corvallis, Oregon 97331

Values below a specified detection limit are a common occurrence in environmental measurement which complicates statistical analysis. This paper addresses the problem of how these data may be included in hypothesis testing and demonstrates a solution that uses a regression model in the SAS statistical package with the ability to accommodate left-censored data. To illustrate, the LIFEREG procedure is applied to data for an analyte (tin) found in fish tissue during a multifactorial bioaccumulation study involving two experimental sites, three experiments at each site, and sampling on multiple days. The results of this analysis are presented, indicating evidence of bioaccumulation but no significant differences between sites or experiments. While regression methods to accommodate left-censored data are not new, their use in environmental studies does not appear to be widely practiced. The strengths and limitations of this approach are discussed.

Introduction

In the measurement of contaminants in water, air, soil, tissue, and other samples, values below a specified detection limit (DL) are a common occurrence which complicates statistical analysis. Often the objective of the measurements is to help distinguish between contaminant levels at two points in time or in two environments that may or may not be different. When the data include many values below the DL, the question of how best to deal with these samples becomes a significant issue with legal and economic implications. For example, one may be required to decide whether a polluted area is cleaner (or more contaminated) than it was before, or whether contaminant levels differ significantly between an area of interest and a reference area designated as unpolluted. Porter et al. (*1*) provide a good general summary of the issues to be considered when faced with many observations below the DL. However, few resources that we found provide specific solutions that can be readily applied to hypothesis testing.

Many common practices for handling values below DL, such as deletion of these samples or simple substitution of the DL, one half the DL or zero for each value below DL, are known to produce biased estimates for means and variances (*2–5*) and also lack sound statistical justification. For example, substituting one-half the DL assumes a uniform distribution between zero and the DL, an assumption which is inconsistent with typical distributional assumptions for values above the DL such as the normal or log-normal. Moreover, replacing each value below the DL with the same value alters the underlying variation attributable to the sample.

When data treated in this manner are used in two-sample hypothesis testing and regression, results can vary widely depending on the substituted value (*6*). Although more sophisticated substitution methods have been proposed for estimating means and variances in water and air samples (*2–4, 7, 8*), methods for estimation and hypothesis testing in regression and analysis of variance models have received less attention. Faced with analyzing a complex experimental design or observational study, it is tempting to use simple substitution and to proceed with the analysis.

The problem of incomplete samples is analogous to that found with survival or failure data, in which for some observations the exact starting time may not be known. This is referred to as left censoring. When the point of censoring is known for all observations and the number of censored observations varies from sample to sample, type I censoring is present. The DL in environmental samples leads to type I left-censored observations.

While regression methods to accommodate left-censored data are not new (*9, 6*), their use in environmental studies does not appear to be widely practiced. The purpose of this paper is to demonstrate the use of these methods for analyzing data from an experimental design using a readily available statistical package. This paper focuses on regression models for handling type I left-censored data. The aim is to show the utility of this approach in which all available data may be used without fabricating values below the DL. As an example, this approach will be taken in a re-analysis of data from a previous study on contaminants in fish tissue in which many analytes were found in samples at levels below the DL.

Experimental Section

Sample Data. The sample data described in this paper are from a larger study called the San Diego Health Effects Study. The purpose of the study was to evaluate the effectiveness of a state-of-the-art wastewater reclamation process as a means of supplementing the municipal potable water supply. Highly treated municipal sewage effluent (AWT water) was compared with an existing raw water source supplying the City of San Diego (Miramar water). Extensive analyses were conducted on both water sources, indicating the presence of low levels of many chemical contaminants below limits of detection. As part of the chemical evaluation, levels of pollutants of interest were measured in fathead minnows maintained in aquaria supplied with the two water sources (*10*). Three 28-day bioaccumulation experiments were conducted during which quadruplicate fish samples were collected on days 14 and 28 after the fish were introduced into the two-source aquaria and also on day 7 in the second and third

* Author for correspondence; e-mail address: pubhr1057@ucsvax.sdsu.edu.

0013-936X/94/0928-0898$04.50/0

experiments on each source. Fish tissue samples were analyzed for organics and inorganics. Most organic analytes could not be recovered from fish tissue at levels in excess of detection limits with sufficiently high frequency to allow any rigorous statistical comparison of the two water types. On the other hand, a number of inorganics were recovered at or above DL with sufficient frequency to warrant statistical analysis. This report will focus on how an existing statistical program (SAS) was adapted for analysis of one such analyte (tin). Of the 64 observations for tin concentration in fish tissue, 20 (31%) fell below the detection limit of 0.067 mg/kg.

Description of Statistical Model. Let y denote the dependent variable (in this example, tin concentration in fish tissue), and let y_o denote a specific value representing the censoring point (the detection limit) such that if $y \geq y_o$, the response is measurable. If $y < y_o$, it is censored (below the DL). y is related to the independent variables, x_i, where $i = 1, ..., k$ by the model

$$y = \sum_{i=1}^{k} \beta_i x_i + \sigma e$$

The β_i's are the regression coefficients to be estimated, σ is a scale parameter, and e is an error term assumed to have a known distribution such as a normal. Let F represent the cumulative distribution function and f the density function of this distribution where for some value t, $F(t) = \text{prob}(e \leq t)$ and $f(t) = dF(t)/dt$. Estimation of the β's is by maximum likelihood. For the jth observation ($j = 1, ..., n$), let

$$w_j = (y_j - \sum_{i=1}^{k} \beta_i x_{ij})/\sigma$$

The log likelihood function to accommodate left-censored data is given by

$$L = \sum_{j=1}^{n} d_j \log(f(w_j)/\sigma) + (1 - d_j) \log(F(w_j))$$

where d_j is an indicator variable which equals 1 if the observation is above the DL and 0 if it is below the DL. Since for type I censoring the value y_o is known, y_o replaces y_j in L (in w_j) for left-censored observations.

The assumed underlying distribution may take a variety of forms; however, for environmental data the normal or log-normal are common choices.

The statistical software to implement the approach described below is readily available in the SAS statistical package through the LIFEREG procedure (*11*). This procedure was designed to analyze survival data using a variety of parametric regression models where censoring may occur on the right or left. The option accommodating left-censored data is not routinely available in survival analysis procedures. Hence, the LIFEREG procedure is somewhat unusual in this regard.

The β's are estimated by maximum likelihood using a Newton–Raphson algorithm. The negative of the matrix of second partial derivatives of L, called the observed information matrix, is computed. The inverse of the matrix yields the covariance matrix from which standard error estimates of the estimated β's are obtained. Details are provided in ref 11.

Given the powerful data manipulation capabilities of the SAS system, substantial flexibility is available with this package in analyzing environmental data. Although LIFEREG is designed as a regression program, data collected from multifactorial experimental designs can, in many cases, be analyzed with this procedure with just a few program instructions. Application to tin concentrations found in fish tissue during the Health Effects Study will provide an illustration.

Table 1. Means and Standard Deviations of Tin Concentrations Found in Fish Exposed to Test Waters Shown by Experiment, Number of Days Exposed, and Source of Test Water

	no. of fish samples	no. of samples below DL (0.067 mg/kg)	mean (mg/kg)	standard deviation
experiment				
1	16	8	0.044	0.106
2	24	8	0.081	0.050
3	24	4	0.093	0.024
no. of days exposed				
7	16	7	0.074	0.032
14	24	8	0.076	0.027
28	24	5	0.100	0.072
source[a]				
AWT	32	9	0.085	0.044
Miramar	32	11	0.075	0.066
total	64	20	0.081	0.056

[a] AWT: advanced wastewater treatment facility effluent. Miramar: Raw water entering municipal treatment facility.

Results

Three questions were addressed by this analysis: (1) Are there significant differences in tin concentrations in fish tissue from the two water sources? (2) Are there differences in levels found in the three separate experiments? and (3) Is there evidence of changing concentrations of this contaminant in fish tissue over time (i.e., bioaccumulation) in a given experiment? The statistical approach to this analysis takes into account the water source from which fish were taken (two levels), the experiment (three levels), and the sampling day (three levels at days 7, 14, and 28) in an analysis of variance.

Table1 displays the means and standard deviations of tin concentration found in fish tissue by the levels of each of the three factors. Means and standard deviations shown in this table were based on all samples, including those below the DL. The data were stratified, and for each level, a model was fitted which included only the grand mean and the parameter σ, yielding maximum likelihood estimates of the mean and standard deviation, respectively, for normally distributed data. The last line of the table shows the overall mean and standard deviation for all samples.

The first model fitted assumes that the data are normally distributed and that the main effects of experiment, day, and source are included as categorical (or class) variables. When only main effects are included, the LIFEREG procedure can be set up to generate the indicator (or dummy) variables required to represent the categorical variables using the CLASS statement. Figure 1 displays the necessary SAS statements, and Table 2 provides the complete set of data.

Relevant details from the output of LIFEREG for this model are displayed in Table 3. Regression estimates and their standard errors for the indicator variables of ex-

Table 2. Data Set BIOMON1.DAT Consisting of Four Variables and 64 Observations

expt no.	day sample taken	source of water[a]	tin concn. in fish tissue[b]	expt no.	day sample taken	source of water[a]	tin concn. in fish tissue[b]
1	14	43	0.068	1	14	44	0
1	14	43	0	1	14	44	0
1	14	43	0.099	1	14	44	0
1	14	43	0	1	14	44	0
1	28	43	0.068	1	28	44	0.078
1	28	43	0.084	1	28	44	0.348
1	28	43	0	1	28	44	0.097
1	28	43	0.080	1	28	44	0
2	7	43	0	2	7	44	0
2	7	43	0.068	2	7	44	0.096
2	7	43	0	2	7	44	0.090
2	7	43	0	2	7	44	0.073
2	14	43	0.119	2	14	44	0.081
2	14	43	0.138	2	14	44	0.091
2	14	43	0.070	2	14	44	0
2	14	43	0.076	2	14	44	0
2	28	43	0.090	2	28	44	0
2	28	43	0.230	2	28	44	0
2	28	43	0.140	2	28	44	0.098
2	28	43	0.130	2	28	44	0.097
3	7	43	0.105	3	7	44	0.099
3	7	43	0	3	7	44	0.130
3	7	43	0	3	7	44	0.114
3	7	43	0.095	3	7	44	0
3	14	43	0.101	3	14	44	0.068
3	14	43	0.068	3	14	44	0.084
3	14	43	0.106	3	14	44	0.080
3	14	43	0.114	3	14	44	0.078
3	28	43	0.097	3	28	44	0.132
3	28	43	0.125	3	28	44	0.092
3	28	43	0.118	3	28	44	0.096
3	28	43	0	3	28	44	0.104

[a] 43 = AWT; 44 = Miramar. [b] 0 indicates value less than DL.

```
DATA MAIN;

/* The input data file consists of 4 variables: EXP is the experiment
   number (1, 2, 3); DAY is the number of days after the experiment
   begins that the sample was taken (7, 14, 28); SOURCE gives the code
   number for water location of fish sampled (43=AWT, 44=Miramar); and
   TIN gives the contaminant value. If the contaminant value is less
   than the DL, the recorded value is 0. */

   INFILE 'C:BIOMON1.DAT';
   INPUT EXP DAY SOURCE TIN;

/* If the contaminant value is less than the DL of .067,
   set LOWER equal to the missing value code (.). Otherwise, LOWER
   takes on the contaminant value above the DL */

   IF TIN < .067 THEN LOWER=.; ELSE LOWER=TIN;

/* If the contaminant value is below the DL, set TIN equal to the DL */

   IF TIN=0 THEN TIN=.067;

PROC LIFEREG;

/* Declare EXP, DAY and SOURCE as class or categorical variables */

   CLASS EXP DAY SOURCE;

/* The left side of the equal sign in the model statment lists two
   variables, LOWER and TIN. If an observation is above the DL,
   LOWER and TIN have the same value which is interpreted as an
   uncensored observation. If an observation is below the DL, LOWER
   is a missing value and TIN equals the DL; this is interpreted
   as a left-censored observation.

   On the right side of the equal sign, the independent variables
   are listed. In this example, the 3 main effects are given. The
   DIST option gives the assumed underlying distribution: NORMAL
   for the normal distribution, and LNORMAL for the lognormal. */

   MODEL (LOWER, TIN) = EXP DAY SOURCE/ DIST=NORMAL;

RUN;
```

Figure 1. SAS program instructions to fit regression model for tin with main effects of experiment, day, and source used to produce output for the results shown in Table 3.

periment, day, and source are presented. In addition, test statistics and corresponding p-values are given for each main effect as well as the individual indicator variables. The tests are based on the inverse of the observed information matrix. For tests involving only one parameter, the form of the statistic is the square of the ratio of the estimated regression parameter divided by its standard error. The statistic follows approximately a χ^2 distribution with 1° of freedom.

Table 3. Regression Model for Tin Found in Fish with Main Effects of Experiment, Day, and Source Included in the Model as Indicator Variables Assuming a Normal Distribution[a]

variable	regression coeff	standard error	χ^2	deg of freedom	p-value
experiment			2.35	2	0.308
1	−0.0294	0.0192	2.35	1	0.125
2	−0.0091	0.0157	0.34	1	0.561
3[b]	–	–	–	–	–
day			7.93	2	0.019
7	−0.0464	0.0189	6.00	1	0.014
14	−0.0365	0.0159	5.27	1	0.022
28[b]	–	–	–	–	–
source			0.08	1	0.78
AWT	0.0039	0.0139	0.08	1	0.78
Miramar[b]	–	–	–	–	–

[a] Log-likelihood = 53.33. [b] Denotes reference category in regression model.

Table 3 indicates that there are differences among sampling days (p = 0.019). This is primarily attributed to higher tin concentrations on day 28 relative to days 7 and 14, indicating evidence of bioaccumulation. The negative coefficients for the dummy variables of days 7 and 14 indicate that the contaminant levels are lower at these periods compared to day 28. This is consistent with the lower means seen in fish tissues on days 7 and 14 compared with day 28 in Table 1. This analysis also indicated that there were no significant differences seen among experiments (p = 0.308). Lastly, no significant differences between the two sources were detected with

Table 4. Fitted Models and Tests of Hypotheses for Tin in Fish Samples

model	log-likelihood
1. main effects (experiment, day, source)	53.329
2. main effects (experiment, source)	49.473
3. main effects (day, source)	52.118
4. main effects (experiment, day)	53.290
5. main effects (experiment, day, source) + day-by-source interaction	55.633

effect tested	model comparison	χ^2 [a]	deg of freedom	p-value (LRT)[b]	p-value (IM)[c]
day main effect	1 vs 2	7.71	2	0.021	0.019
expt main effect	1 vs 3	2.42	2	0.298	0.308
source main effect	1 vs 4	0.08	1	0.780	0.781
day-by-source int.	5 vs 1	4.61	2	0.10	–

[a] Likelihood ratio test statistic. Obtained by taking two times the difference in log-likelihoods between the two models being compared. For example, to compare models 1 and 2: χ^2 statistic = 2 × (53.329 − 49.473) = 7.72. [b] Corresponding p-value for the likelihood ratio test. [c] Corresponding p-value for the test based on the information matrix. Included for comparison purposes. Same p-values as in Table 3.

regard to tin levels found in fish maintained in these sources (p = 0.781).

For small samples, likelihood ratio tests may be preferable. This may be accomplished by fitting several models and comparing full with restricted models by taking two times the difference in the log-likelihoods. LIFEREG prints the log-likelihood for each model. The test statistic follows approximately a χ^2 distribution where the number of degrees of freedom is the difference in the number of regression parameters estimated between the two models.

Table 4 displays several likelihood ratio tests for the tin concentration data along with the tests based on the information matrix which were previously described. The first part of Table 4 shows the log-likelihoods for five fitted models. To construct a likelihood ratio test for the effect of day adjusting for experiment and source, model 1 is compared to model 2. The effects in model 2 are a subset of those in model 1. As shown in Table 4, two times the difference between the full (model 1) and restricted (model 2) log-likelihoods determines the test statistic. There are 2° of freedom since two indicator variables are required to represent the day main effect. Again, the test suggests differences among days (p = 0.021). Comparing the two sets of p-values in Table 4 suggests close agreement between the two types of tests. With smaller samples or higher censoring rates, some divergence may occur.

The first four models in Table 4 only include subsets of the main effects. In factorial designs, an investigator may be interested in assessing first-order or even higher-order interactions. In some studies, assessment of interactions may be the main focus of the analyses. Although LIFEREG is designed primarily to assess main effects, interactions may be included by constructing the necessary indicator variables to represent them in a previous step. As an example, we describe how to include the day-by-source interaction into a model and how to test for it using the likelihood ratio method. Figure 2 displays the SAS program instructions to fit a model which includes the main effects and this interaction. Note that the analyst must set up a dummy variable scheme in the DATA step to represent the categories for day and source. In addition, two more dummy variables are created by taking the

```
DATA MAIN;

/* Descriptions of the instructions for the following 4 statements
   are given in Figure 1. */

   INFILE 'C:BIOMON1.DAT';
   INPUT EXP DAY SOURCE TIN;
   IF TIN < .067 THEN LOWER=.; ELSE LOWER=TIN;
   IF TIN=0 THEN TIN=.067;

/* Two indicator variables, D_EX1 and D_EX2, for Experiment are
   created.  First, the variables are initialized to 0.  Then if
   the incoming observation is from Experiment 1, change D_EX1 to 1.
   If it is from Experiment 2, change D_EX2 to 1.  Experiment 3 is
   the reference category represented by both indicator variables
   equaling 0.  */

   D_EX1=0; D_EX2=0;
   IF EXP=1 THEN D_EX1=1;
   ELSE IF EXP=2 THEN D_EX2=1;

/* In a similar way, the indicator variables D_DY1 and D_DY2 are
   created to represent Day.  Day 7 is represented by D_DY1 = 1 and
   D_DY2 = 0; Day 14 represented by D_DY1 = 0 and D_DY2 = 1; and
   Day 28 is the reference category in which both variables are 0. */

   D_DY1=0; D_DY2=0;
   IF DAY=7 THEN D_DY1=1;
   ELSE IF DAY=14 THEN D_DY2=1;

/* D_SRCE is the indicator variable for Source.  If the incoming
   observation is from AWT (43), then D_SRCE is set to 1.  Otherwise,
   it remains at its initialized value of 0 to represent Miramar (44)
   */

   D_SRCE=0;
   IF SOURCE=43 THEN D_SRCE=1;

/* The next 2 statements create the indicator variables to represent
   the Day by Source interaction.  The indicator variable for Source
   is multiplied by each of the indicator variables for Day to create
   DAYSRCE1 and DAYSRCE2.  In a similar way other interaction terms
   could be created. */

   DAYSRCE1 = D_DY1*D_SRCE;
   DAYSRCE2 = D_DY2*D_SRCE;

PROC LIFEREG;

/* The left side of the equal sign in the MODEL statement was
   described in Figure 1.      The right side lists all of the
   indicator variables to represent Experiment, Day, Source, and
   the Day by Source interaction. */

MODEL (LOWER, TIN) = D_EX1 D_EX2 D_DY1 D_DY2 D_SRCE
                     DAYSRCE1 DAYSRCE2 / DIST=NORMAL;
RUN;
```

Figure 2. SAS program instructions to fit regression model for tin with main effects of experiment, day, and source and the day-by-source interaction using indicator variables.

product of the term for source with each of the two terms representing day; these variables represent the interaction. In the MODEL statement, all of the dummy variables representing the main effects and the interaction are listed. Model 5 in Table 4 shows the likelihood information after the fit. By constructing a likelihood ratio test to compare models 5 and 1, a test of the day-by-source interaction is obtained (p = 0.10). This approach can be expanded to include all interaction terms of interest.

The normal distribution was selected as the underlying distribution for the examples in this report. As mentioned earlier, other distributions could be selected, notably the log-normal. This is easily accomplished by changing the option DIST=NORMAL to DIST=LNORMAL in the MODEL statement. The choice of which distribution to use will likely depend upon initial descriptive statistics, plots suggesting extreme skewness, or knowledge about the underlying biologic process. For data above the DL, the log transform did not dramatically change the shape of the distribution for tin. Repeating the analysis using transformed data does not test for distributional assumptions. If similar conclusions are reached, it may lend weight to those conclusions; but the choice of which distribution to use is not answered by the re-analysis. However, to further illustrate the technique, the main effects model with a log-normal distribution was fitted. Again, the water source was not statistically significant (p = 0.604). Day was significant as before, although the p-value (p = 0.008) was somewhat smaller. Experiment was marginally significant (p = 0.054); the difference between experiments 1 and 3 was the largest contributor, where experiment 1 had lower levels than experiment 3 on the log scale.

Discussion

This paper provides an example of how values below a detection limit can be incorporated into a multifactorial analysis without arbitrarily assigning a fixed value to these quantities such as zero or one-half the DL. The method borrows from the survival or failure-time problem known as type I left censoring. In constructing a likelihood, if a value is above the DL, the contribution to the likelihood is the probability density function. An observation below the DL contributes the probability from the cumulative distribution function evaluated at the DL. The LIFEREG procedure of SAS provides the existing software to implement this method. While the emphasis in this report has been on the analysis of a factorial design, the regression format of this procedure allows for a variety of multivariable problems for both experimental and observational studies. Both categorical and continuous variables may be incorporated as independent variables. We demonstrated that interaction terms may also be included in the model with little additional effort.

This regression model has been used in econometrics for some time (*9*) and is referred to as "tobit" analysis. Clearly, this model may be used for two-sample problems as noted by Helsel (*6*). A single indicator variable is included as the only independent variable denoting group identification as zero or one. A test of the regression coefficient is also a test of the difference in group means. However, one may prefer to use nonparametric methods such as the Wilcoxon rank sum test and avoid the distributional assumptions, although there may be some loss of power.

Although not demonstrated in this paper, multiple detection limits can be accommodated using tobit analysis. This may occur when different DL's are used among subsets of observations in a sample (*4, 6*). Data combined from several laboratories or where the DL changes over time are examples.

We found little evidence of use of this approach in the environmental science literature. As part of their study, Vance et al. (*12*) used this method to analyze trace elements in fingernail and hair samples. Although a brief description is given of the statistical method, the paper was not intended as a comprehensive discussion on the methodology and its implementation.

In our example, 31% of the observations fell below the detection limit out of a total sample of 64. As the percent of observations below the DL increases, bias in the maximum likelihood estimates of means, standard deviations, and regression parameters is likely to increase. Further work is required to evaluate the degree of bias. In addition, since sample sizes for environmental studies can be rather small, further work is needed to evaluate the small-sample performance of the test statistics used in this paper.

Some data analysts may prefer to use nonparametric methods for fitting left-censored data to avoid distributional assumptions and hypothesis testing based on large-sample theory. Nonparametric tests are widely available for the two-sample problem. However, nonparametric regression or ranking procedures for multifactorial designs are not readily available. Consistent with good statistical practice, the analyst should check assumptions based on the observed data, as well as be aware of the effects of outliers and multicollinearity. Graphical analysis of residuals is always recommended; however with censored data, the use of residuals is complicated since the set of residuals is incomplete.

The LIFEREG procedure in SAS is inappropriate for handling more complex problems. Repeated measure designs in which multiple observations are taken on the same sampling unit require models which account for the error variability within sampling units and between units. LIFEREG does not handle these types of problems. Similarly, longitudinal studies where multiple observations are made on the same sampling unit over time are not considered.

Our intent has been to demonstrate the utility of this regression method and make it more accessible for the analysis of complex environmental data. Unlike other reports, this discussion has emphasized the analysis of multifactorial studies. However, the regression framework makes it applicable to a wide variety of problems facing environmental scientists and engineers. Advanced programming skills are not required to implement this method, which is available in a widely-used statistical package. We hope that this demonstration will encourage researchers to abandon the deletion and simple substitution methods for handling data below detection limits in favor of a more comprehensive and statistically sound solution.

Literature Cited

(1) Porter, P. S.; Ward, R. C.; Bell, H. F. *Environ. Sci. Technol.* **1988**, *22*, 856–861.
(2) Gilliom, R. J.; Helsel, D. R. *Water Resour. Res.* **1986**, *22*, 135–155.
(3) Gleit, A. *Environ. Sci. Technol.* **1986**, *19*, 1201–1206.
(4) Newman, M. C.; Dixon, P. M.; Looney, B. B.; Pinder, J. E. *Water Resour. Bull.* **1989**, *25*, 905–915.
(5) ASTM Subcommittee D19.02. *Annual Book of ASTM Standards*; American Society for Testing and Materials: Philadelphia, 1992; Chapter D, 11.01, pp 14–20.
(6) Helsel, D. R. *Environ. Sci. Technol.* **1990**, *24*, 1766–1774.
(7) Rao, S. T.; Ku, J. Y.; Rao, K. S. *J. Air Waste Manage. Assoc.* **1991**, *41*, 442–448.
(8) Cohen, M. A.; Ryan, P. B. *J. Air Waste Manage. Assoc.* **1989**, *39*, 328–329.
(9) Tobin, J. *Econometrica* **1958**, *26*, 24–36.
(10) de Peyster, A.; Donohoe, R.; Slymen, D. J.; Froines, J. R.; Olivieri, A. W.; Eisenberg, D. M. *J. Toxicol. Environ. Health* **1993**, *39*, 121–142.
(11) SAS Institute Inc. *SAS/STAT User's Guide, Version 6*, 4th ed.; SAS Institute Inc.: Cary, NC, 1989; Vol. 2, pp 997–1025.
(12) Vance, D. E.; Ehmann, W. D.; Markesbery, W. R. *Biol. Trace Element Res.* **1988**, *17*, 109–121.

Received for review August 20, 1993. Revised manuscript received January 3, 1994. Accepted January 18, 1994.[⊗]

[⊗] Abstract published in *Advance ACS Abstracts,* March 1, 1994.